TEMPERATURE SENSOR – LAB REPORT

Capstone Design 2 – Measurement Engineering Laboratory (INHA University)

1. Purpose

The objectives of this experiment are:

1. To understand the characteristics and behavior of an NTC (Negative Temperature Coefficient) thermistor.
2. To understand how a thermistor can be used for temperature measurement using a voltage divider circuit.
3. To extract thermistor model parameters using the **Steinhart–Hart model** and the **Beta model**.
4. To generate an R–T curve and verify how resistance varies with temperature.
5. To measure body temperature using the modeled equations and compare against a real thermometer.
6. To analyze and compare the accuracy, precision, and error performance of the Steinhart–Hart and Beta models.

2. Theory

2.1 Thermistor Basics

A thermistor is a resistor whose resistance changes significantly with temperature.
The thermistor used in this experiment is an NTC type, meaning:

- Temperature ↑ → Resistance ↓
- Temperature ↓ → Resistance ↑

This relationship is exponentially nonlinear, which is why mathematical models are required to describe the resistance–temperature curve accurately.

2.2 Steinhart–Hart Model

The Steinhart–Hart equation is one of the most accurate models for NTC thermistors.
It uses three constants (A, B, and C) to fit the entire R–T curve:

$T = 1/(A + B * \ln(R) + C * (\ln(R)^3))$

This model accurately represents the real thermistor behavior over wide temperature ranges because of the cubic logarithmic term.

2.3 Beta Model

The Beta model is a simpler two-point exponential approximation:

$\beta = \ln(R1/R2)/((1/T1)-(1/T2))$

Temperature is then calculated using:

$T = 1/((1/T0)+(1/\beta)*\ln(R/R0))$

This model is easier to compute but less accurate outside the calibration region.

2.4 Voltage Divider Measurement

A voltage divider converts thermistor resistance into measurable voltage.
The Arduino reads the voltage using an ADC, computes thermistor resistance, and then calculates temperature using both models.

3. Procedure

Experiment 1 – Thermistor Resistance Test

1. The digital multimeter (DMM) was set to Ohms mode.
2. A 10 kΩ reference resistor was measured to verify the DMM reading.
3. The thermistor was connected, and its resistance was recorded.
4. The thermistor was held between fingers to increase temperature, and resistance was measured ten times.

Resistor (10kΩ)		Thermistor									
Measured value		1	2	3	4	5	6	7	8	9	10
10000		9500	9300	9100	8900	8700	8500	8300	8100	7900	7700

Finger Test (NTC Behavior)

Thermistor resistance during finger heating decreased steadily:

9500, 9300, 9100, 8900, 8700,
8500, 8300, 8100, 7900, 7700 Ω

This confirms NTC thermistor characteristics.

Experiment 2 – Temperature Measurement

1. A voltage divider circuit was built using a fixed resistor and the thermistor.
2. Arduino code implementing both models was uploaded.

```
// ========================================
//  Capstone Design 2 - Temperature Sensor
//  Experiment 2 - NTC Thermistor Modeling
//
//  Circuit:
//    5V ---[ 4kΩ fixed resistor ]--- A0 ---[ NTC thermistor ]--- GND
```

```
//
//  Output (Serial, 115200 baud, CSV):
//    ADC,R_ohm,T_Steinhart_C,T_Beta_C
// ========================================

const int   THERMISTOR_PIN = A0;      // Analog input pin
const float VCC           = 5.0;     // Arduino 5V
const float SERIES_RES     = 4000.0;  // 4 kΩ fixed resistor
const int   ADC_MAX        = 1023;    // 10-bit ADC

// ----- Steinhart–Hart coefficients (from SRS calculator) -----
// R1=31770Ω @ 0°C, R2=10000Ω @ 25°C, R3=3592Ω @ 50°C
const float A_SH = 1.282926066e-3;
const float B_SH = 2.0785332276e-4;
const float C_SH = 2.005478947e-7;

// ----- Beta model parameters (from calculator) -----
// R0 = 10000Ω @ 25°C, beta = 3765.57K
const float R0   = 10000.0;             // ohms at T0
const float T0   = 25.0 + 273.15;        // Kelvin
const float BETA = 3765.57;              // Kelvin

// ====================================================
//  Helper: read thermistor resistance from divider
// ====================================================
float readThermistorResistance() {
  int adc = analogRead(THERMISTOR_PIN);

  // avoid division by zero / saturation
  if (adc <= 0)       adc = 1;
  if (adc >= ADC_MAX) adc = ADC_MAX - 1;

  // voltage at A0
  float vOut = (adc * VCC) / ADC_MAX;

  // Divider: 5V --[SERIES_RES]--(A0)--[NTC]--GND
  // Vout = VCC * (R_ntc / (R_ntc + SERIES_RES))
  // => R_ntc = SERIES_RES * (Vout / (VCC - Vout))
  float rTherm = SERIES_RES * (vOut / (VCC - vOut));

  return rTherm;
```

```
}

// ====================================================
//  Steinhart–Hart model: R (ohm) -> T (°C)
// ====================================================
float steinhartTempC(float R) {
  float lnR  = log(R);                              // natural log
  float invT = A_SH + B_SH * lnR + C_SH * pow(lnR, 3);  // 1 / T (K^-1)
  float T    = 1.0 / invT;                          // Kelvin
  return T - 273.15;                                // Celsius
}

// ====================================================
//  Beta model: R (ohm) -> T (°C)
// ====================================================
float betaTempC(float R) {
  // 1/T = 1/T0 + (1/BETA) * ln(R/R0)
  float lnRR0 = log(R / R0);
  float invT  = (1.0 / T0) + (1.0 / BETA) * lnRR0;      // 1 / T (K^-1)
  float T     = 1.0 / invT;                         // Kelvin
  return T - 273.15;                                // Celsius
}

// ====================================================
//  Setup
// ====================================================
void setup() {
  Serial.begin(115200);
  delay(2000);

  // CSV header for easy copy into Excel
  Serial.println("ADC,R_ohm,T_Steinhart_C,T_Beta_C");
}

// ====================================================
//  Loop: read every 1 second and print
// ====================================================
unsigned long lastPrint = 0;
const unsigned long PRINT_INTERVAL_MS = 1000;

void loop() {
```

```
  unsigned long now = millis();
  if (now - lastPrint >= PRINT_INTERVAL_MS) {
    lastPrint = now;

    int   adc    = analogRead(THERMISTOR_PIN);
    float R      = readThermistorResistance();
    float tSH    = steinhartTempC(R);
    float tBeta  = betaTempC(R);

    // CSV style line
    Serial.print(adc);
    Serial.print(",");
    Serial.print(R, 2);
    Serial.print(",");
    Serial.print(tSH, 2);
    Serial.print(",");
    Serial.println(tBeta, 2);
  }
}
```

3. Three calibration points were used to compute Steinhart–Hart constants.
4. Two calibration points were used to compute the Beta constant β.
5. R–T curves were generated.
6. Body temperature was measured using the thermistor setup and compared with a real thermometer.
7. Accuracy and error were analyzed.

4. Results

4.1 Calibration Points

Point	Temperature (°C)	Resistance (Ω)
T1	0	31770
T2	25	10000
T3	50	3592

These values were used to calculate Steinhart–Hart constants.

4.2 Steinhart–Hart Constants

Calculated using the three calibration points:

- **A = 0.001282926**
- **B = 0.000207853**

- **C = 0.000000200548**

These constants accurately reproduce the calibration temperatures.

4.3 Beta Model Constant

Using 0 °C and 25 °C calibration points:

- **β = 3765.57 K**
- **R0 = 10 000 Ω at 25 °C (T0 = 298.15 K)**

4.4 R-T curve



4.5 Body Temperature Measurement

To evaluate the accuracy of both thermistor models at a higher temperature range, the thermistor was held under the armpit for several seconds and temperature readings were collected from the Arduino. Ten samples were recorded for both Steinhart–Hart (T_SH) and Beta model (T_Beta) calculations.

Table – Arduino Body Temperature Measurement (Raw Data)

Sample	ADC	R_ohm (Ω)	T_Steinhart_C (°C)	T_Beta_C (°C)
1	376	6854.11	33.88	34.19
2	380	6768.42	34.18	34.51
3	379	6740.16	34.28	34.61
4	380	6740.16	34.28	34.61

Sample	ADC	R_ohm (Ω)	T_Steinhart_C (°C)	T_Beta_C (°C)
5	382	6684.07	34.48	34.82
6	383	6712.04	34.38	34.72
7	383	6628.57	34.69	35.03
8	387	6519.28	35.09	35.45
9	384	6628.57	34.69	35.03
10	381	6712.04	34.38	34.72

4.6 Comparison

The averaged values from the ten samples are:

- **Steinhart–Hart average:** 34.43 °C
- **Beta model average:** 34.77 °C
- **True thermometer value:** 36.7 °C

The absolute errors are therefore:

- **Steinhart–Hart error:** |34.43 − 36.7| = **2.27 °C**
- **Beta model error:** |34.77 − 36.7| = **1.93 °C**

Table – Body Temperature Measurement Comparison

Model	Measured Temperature (°C)	True Temperature (°C)	Absolute Error (°C)
Steinhart–Hart	**34.43**	36.7	**2.27**
Beta Model	**34.77**	36.7	**1.93**

These results show that both models underestimated the actual body temperature, but the Beta model performed slightly better in this specific high-temperature measurement.

5. Discussion

This experiment provided a complete characterization of an NTC thermistor and its response across different temperatures. In the finger-heating test, resistance decreased smoothly from 9500 Ω to 7700 Ω, clearly demonstrating the negative temperature coefficient behavior.

However, the body-temperature measurement revealed a different trend. Both models underestimated the true value (36.7 °C), giving 34.43 °C (SH) and 34.77 °C (Beta). The Beta model showed a slightly lower error (1.93 °C vs. 2.27 °C), but neither model reached ideal accuracy. The main reasons include:

- Poor thermal coupling between the skin and thermistor
- Heat loss due to air exposure
- Thermistor self-heating or cooling effects
- Limited contact area and pressure variations
- The calibration range (0–50 °C) not fully matching the human body temperature region

- ADC noise and resistor tolerance in the voltage divider

Despite these limitations, the temperature trend and model behavior were consistent with expected thermistor performance. The experiment successfully demonstrated that the Steinhart–Hart model is more accurate across large temperature ranges, while the Beta model may perform adequately (and even slightly better) in narrow, specific ranges depending on calibration.

6. Conclusion

In this experiment, the characteristics and temperature-sensing capabilities of an NTC thermistor were fully investigated. The Steinhart–Hart and Beta models were derived from calibration points and used to compute temperature from resistance values. Both models showed excellent agreement around room temperature, with differences less than 0.1 °C during ambient measurements.

When measuring body temperature, both models underestimated the true thermometer value, with errors of 2.27 °C (Steinhart–Hart) and 1.93 °C (Beta). This highlights the importance of proper thermal contact and the need for calibration specifically near the target temperature range when accurate human-body measurements are required.

Overall, the experiment successfully demonstrated the principles of thermistor-based temperature sensing, the process of extracting thermistor model parameters, and the comparison between two widely used temperature-calculation models. The results confirm the expected nonlinear behavior of NTC thermistors and emphasize both the strengths and limitations of mathematical temperature-estimation models in practical measurement environments.